;     Функционално програмиране.

;     Упражнение N4

;                     n       n        n-1    0

;     Пресмятане на  b  .    b  = b . b    , b  = 1

;

;     Линеен рекурсивен процес.

(define (expt1 b n)

   (if (= n 0)

       1

       (* b (expt1 b (- n 1)))))

;     Линеен итерационен процес.

(define (expt2 b n)

   (exp-iter b n 1))

(define (exp-iter b counter product)

   (if (= counter 0)

       product

       (exp-iter b (- counter 1) (* b product))))

;     Логаритмичен рекурсивен процес чрез последователно повдигане

;                       8       2         4     2 2   8     4 2

;на квадрат, например  b  чрез b  = b.b, b  = (b ) , b  = (b )

;        n

;        -

; n      2  2

;b  = (b   )  , n - четно,

;

; n        n-1

;b  = b . b    , n - нечетно.

(define (fast-exp b n)

   (cond ((= n 0) 1)

         ((even? n) (square (fast-exp b (/ n 2))))

         (else (* b (fast-exp b (- n 1))))))

;     Задача 1.Да се напише функция, която реализира

;                                    n

;итеративен процес за пресмятане на b , който използва

;последователно повдигане на квадрат и е логаритмичен.

;Да се използва, че                  n

;                                    -

;                          n      2  2

;                         b  = (b   )  .

;Заедно с b и n да се съхранява и променлива на състоянието a.    n

;Да  се  дефинира  трансформация  на  състоянието  така,  че   a.b

;да не се променя от едно състояние към друго. В началото а е 1 и

;отговорът се дава от стойността на а в края на процеса. Тoва се

;нарича техника на инвариантното количество, което остава

;непроменено от състояние към състояние.

(define (fast-exp1 b n a)

  (cond ((= n 0) a)

        ((even? n) (fast-exp1 (square b) (/ n 2) a))

        (else (fast-exp1 b (- n 1) (* a b)))))

 (define (square x) (* x x))

;     Задача 2. Следващата функция реализира умножение

;чрез събиране и описва линеен рекурсивен процес.

;(define (* a b)

;   (if (= b 0)

;       0

;       (+ a (* a (- b 1)))))

;     Чрез функциите double, която удвоява цяло число, и halve,

;която дели четно число на 2, да се напише функция за умножение,

;която работи в логаритмично време и реализира рекурсивен процес.

(define (halve n) (/ n 2))

(define (double n) (+ n n))

(define (fast-mult a b)

  (cond ((= b 0) 0)

        ((even? b) (double (fast-mult a (halve b))))

        (else (+ a (fast-mult a (- b 1))))))

;     Задача 3. Използвайки резултатите от 1 и 2 задача

;да се напише функция, която генерира итерационен процес за

;умножение на две числа чрез събиране, удвояване на число,

;разполовяване на четно число и работи в логаритмично време.

(define (mult-iter a b product)

  (cond ((= b 0) product)

        ((even? b) (mult-iter (double a) (halve b) product))

        (else (mult-iter a (- b 1) (+ product a)))))

;(mult-iter a b 0)

;     Задача 4. Да се напише функция, която генерира линеен

;итеративен процес за събиране на две положителни цели числа, 

;използвайки функцията ADD1, която прибавя 1 към числовия си 

;аргумент, и SUB1, която изважда 1 от числовия си аргумент.

(define (pl a b)

  (if (= b 0) 

      a

      (pl (add1 a) (sub1 b))))

